
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk25 related to the title of this CR.

Contact:
Shaun Harry

shaun.harry@keysight.com
2 Corrections required
2.1 Change 1
	Function name
	fl_TC_8_1_5_5_1_NR5GC_EUTRA_TestBody()

	Reason for change
	Trigger EUTRA->NR after step 1 is not needed. NR PTC will anyway stop inside f_NR5GC_IRAT_S1ToN1_IdleMode() to receive GUTI paramters from EUTRA

	Summary of change
	Remove wait for trigger

	TTCN module
	RRC_Others_NR5GC_EUTRA

	MCC160 Comment
	


Before Change

	function fl_TC_8_1_5_5_1_TestBody() runs on NR5GC_PTC

  {

    //Switch on NR Cell 1

    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2); // FFS: This is missing in the prose

    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_Trigger);

    // @siclog "Step 1" siclog@

    // Takes place in EUTRA PTC

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);
    //@siclog "Step 2" siclog@

    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.9 indicate that the UE has performed mobility registration updating when it camps on NR Cell 1?

    f_NR5GC_IRAT_S1ToN1_IdleMode(nr_Cell1);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

  }


After Change

	  function fl_TC_8_1_5_5_1_TestBody() runs on NR5GC_PTC

  {

    //Switch on NR Cell 1

    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2); // FFS: This is missing in the prose

    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_Trigger);

    // @siclog "Step 1" siclog@

    // Takes place in EUTRA PTC

    //REMOVED f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);
    //@siclog "Step 2" siclog@

    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.9 indicate that the UE has performed mobility registration updating when it camps on NR Cell 1?

    f_NR5GC_IRAT_S1ToN1_IdleMode(nr_Cell1);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

  }




2.2 Change 2
	Function name
	fl_TC_8_1_5_5_1_NR5GC_EUTRA_TestBody

	Reason for change
	1. Trigger EUTRA->NR after step 1 is not needed.

2. EUTRA PTC should not delay after connection release in this case, because the UE may try to access the NR cell immediately. 

	Summary of change
	1. Remove trigger to NR
2. Pass parameter into f_EUTRA_RRC_ConnectionRelease_Common to not start the delay timer, see change 3
Note : Even following these changes, there is still a potential race condition because the NR PTC is not ready to receive RRCSetupRequest until it has received the GUTI parameters from EUTRA cell, which are only sent on return from  f_EUTRA_RRC_ConnectionRelease_Common() (at the beginning of f_EUTRA38_NR5GC_S1ToN1_IdleMode()) 

	TTCN module
	RRC_Others_NR5GC_EUTRA

	MCC160 Comment
	


Before Change

	function fl_TC_8_1_5_5_1_NR5GC_EUTRA_TestBody(EUTRA_NR_CoOrd_SysInfo_Type p_CoOrd_SysInfo) runs on EUTRA_5GS_PTC

  {

    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo := cs_RedirectionInfo_Nr(p_CoOrd_SysInfo.NR[0].Arfcn,

                                                                                                f_NR_ConvertIntToSCS_CarrierInfoNR(p_CoOrd_SysInfo.NR[0].SCS)); //SubcarrierSpacingSSB is FFS

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR);

    // @siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionRelease message indicating redirection to NR Cell 1.

    f_EUTRA_RRC_ConnectionRelease_Common(eutra_Cell1, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other, v_RedirectedCarrierInfo), cs_TimingInfo_Now);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);
   //@siclog "Step 2" siclog@

    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.9 indicate that the UE has performed mobility registration updating when it camps on NR Cell 1?

    f_EUTRA38_NR5GC_S1ToN1_IdleMode(eutra_Cell1);

  }


After Change

	  function fl_TC_8_1_5_5_1_NR5GC_EUTRA_TestBody(EUTRA_NR_CoOrd_SysInfo_Type p_CoOrd_SysInfo) runs on EUTRA_5GS_PTC

  {

    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo := cs_RedirectionInfo_Nr(p_CoOrd_SysInfo.NR[0].Arfcn,

                                                                                                f_NR_ConvertIntToSCS_CarrierInfoNR(p_CoOrd_SysInfo.NR[0].SCS)); //SubcarrierSpacingSSB is FFS

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR);

    // @siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionRelease message indicating redirection to NR Cell 1.

    f_EUTRA_RRC_ConnectionRelease_Common(eutra_Cell1, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other, v_RedirectedCarrierInfo), cs_TimingInfo_Now, false);

    //REMOVED f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);
   //@siclog "Step 2" siclog@

    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.9 indicate that the UE has performed mobility registration updating when it camps on NR Cell 1?

    f_EUTRA38_NR5GC_S1ToN1_IdleMode(eutra_Cell1);

  }


2.3 Change 3
	Function name
	f_EUTRA_RRC_ConnectionRelease_Common

	Reason for change
	Currently on EUTRA RRC connection release, the TTCN will apply a delay (tsc_EUTRA_DelayAfterRRCConnectionRelease ) to allow the UE to complete cell selection before performing any further action. In this redirection case, the UE may try to access NR cell while this timer is running and will not be ready for this until GUTI information has been received.

	Summary of change
	Add parameter to disable the delay time following RRCConnectionRelease

	TTCN module
	EUTRA_RRCSteps

	MCC160 Comment
	


Before Change

	  function f_EUTRA_RRC_ConnectionRelease_Common(EUTRA_CellId_Type p_CellId,

                                                template (value) DL_DCCH_Message p_RRCConnectionRelease,

                                                template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now) runs on EUTRA_PTC

  { /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    /* @sic R5-123740 change 2: additional delay of 640ms at the end of the generic RRC Connection Release procedure (result of AP#55.04, replacing R5-1421779 change 14) sic@ */

    /* @sic R5s150338: p_DrbConfigList removed sic@ */

    var SubFrameTiming_Type v_TimingAtT;   // time T: sending of RRCConnectionRelease acc. to 36.523-3 clause 7.18

    var SubFrameTiming_Type v_TimingNow;

    var integer v_MilliSecondsDelay := tsc_EUTRA_DelayBeforeIntraCellHO;

    var integer v_Duration;

    var SysinfoType_Type v_SysinfoType := f_EUTRA_CellInfo_GetSysinfoType(p_CellId); //@sic R5-173005 sic@

    timer t_Timer;

    if (f_TimingInfo_IsNow(p_TimingInfo)) {

      v_TimingAtT := f_EUTRA_GetNextSendOccasion(p_CellId, v_MilliSecondsDelay);

      v_Duration := v_MilliSecondsDelay;

    }

    else {

      // dedicated timing info is given

      v_TimingAtT := valueof(p_TimingInfo.SubFrame);

      v_TimingNow := f_EUTRA_GetCurrentTiming(p_CellId);

      v_Duration := f_SubFrameTiming_Duration(v_TimingNow, v_TimingAtT);

    }

    if (v_SysinfoType==sysinfoOnly) { //Acc to 36.508 cl 7.18

      v_Duration := v_Duration + tsc_EUTRA_DelayAfterRRCConnectionRelease;   // @sic R5s120683 sic@

    } else { // BL/CE UE acc to 36.523-3 cl 7.18A step 7 @sic R5-173005 sic@

      // 7.   Delay of 1880ms(1040 +840)

      // 1880 ms is 1.25s - 210ms => As the delay is from T, the delay is set to 1250 ms

      v_Duration := v_Duration + tsc_EUTRA_BL_CE_DelayAfterRRCConnectionRelease;

    }

    t_Timer.start (int2float(v_Duration) / 1000.0);

    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,

                                 cs_TimingInfo_EUTRA_NB(v_TimingAtT),

                                 p_RRCConnectionRelease));

    f_EUTRA_RRC_ConnectionRelease_Local(p_CellId, v_TimingAtT);

    alt {

      [] t_Timer.timeout {}

      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {

        // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a sub-sequent function

        t_Timer.stop;

      }

    }

  }


After Change

	function f_EUTRA_RRC_ConnectionRelease_Common(EUTRA_CellId_Type p_CellId,

                                                template (value) DL_DCCH_Message p_RRCConnectionRelease,

                                                template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                                boolean p_Wait := true) runs on EUTRA_PTC

  { /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    /* @sic R5-123740 change 2: additional delay of 640ms at the end of the generic RRC Connection Release procedure (result of AP#55.04, replacing R5-1421779 change 14) sic@ */

    /* @sic R5s150338: p_DrbConfigList removed sic@ */

    var SubFrameTiming_Type v_TimingAtT;   // time T: sending of RRCConnectionRelease acc. to 36.523-3 clause 7.18

    var SubFrameTiming_Type v_TimingNow;

    var integer v_MilliSecondsDelay := tsc_EUTRA_DelayBeforeIntraCellHO;

    var integer v_Duration;

    var SysinfoType_Type v_SysinfoType := f_EUTRA_CellInfo_GetSysinfoType(p_CellId); //@sic R5-173005 sic@

    timer t_Timer;

    if (f_TimingInfo_IsNow(p_TimingInfo)) {

      v_TimingAtT := f_EUTRA_GetNextSendOccasion(p_CellId, v_MilliSecondsDelay);

      v_Duration := v_MilliSecondsDelay;

    }

    else {

      // dedicated timing info is given

      v_TimingAtT := valueof(p_TimingInfo.SubFrame);

      v_TimingNow := f_EUTRA_GetCurrentTiming(p_CellId);

      v_Duration := f_SubFrameTiming_Duration(v_TimingNow, v_TimingAtT);

    }

    if (v_SysinfoType==sysinfoOnly) { //Acc to 36.508 cl 7.18

      v_Duration := v_Duration + tsc_EUTRA_DelayAfterRRCConnectionRelease;   // @sic R5s120683 sic@

    } else { // BL/CE UE acc to 36.523-3 cl 7.18A step 7 @sic R5-173005 sic@

      // 7.   Delay of 1880ms(1040 +840)

      // 1880 ms is 1.25s - 210ms => As the delay is from T, the delay is set to 1250 ms

      v_Duration := v_Duration + tsc_EUTRA_BL_CE_DelayAfterRRCConnectionRelease;

    }

    if (p_Wait) {
      t_Timer.start (int2float(v_Duration) / 1000.0);

    }
    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,

                                 cs_TimingInfo_EUTRA_NB(v_TimingAtT),

                                 p_RRCConnectionRelease));

    f_EUTRA_RRC_ConnectionRelease_Local(p_CellId, v_TimingAtT);

    if (p_Wait) {

    alt {


      [] t_Timer.timeout {}


      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {


        // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a sub-sequent function


        t_Timer.stop;


      }


    }
    }

  }


